# 第3章 重回帰分析

【 課題 】 教科書 108-109 ページの例 5 のデータ (表 3-3) について、 TSP を用いて重回帰分析をおこなってみる。 相関係数行列と偏相関係数行列を求める。 という 2 つのことをおこなってみよう。

### <u>1. Excelデータの作成</u>

分析の準備として、Excelのデータファイルを作成しておく。ここでは、ex3-3.xls という名前 で、Excel4.0 ワークシート形式で保存する。

なお、家計消費支出の変数名を CE に変更してある。これは、C という変数名が TSP では定数 項を表すため、読みこみに際してエラーとなるからである。

	A	В	С	D
1	Date	CE	YD	YD1
2	1983	185234.0	219047.0	214658.4
3	1984	189610.5	223793.4	219047.0
4	1985	197124.2	230667.3	223793.4
5	1986	203577.3	236083.4	230667.3
6	1987	211994.4	240775.4	236083.4
7	1988	222311.8	253954.6	240775.4
8	1989	233017.5	265997.6	253954.6
9	1990	243711.2	279366.1	265997.6
10	1991	250592.4	291102.9	279366.1
11	1992	256900.5	295572.9	291102.9
12	1993	260282.1	297769.1	295572.9
13	1994	267276.0	302224.7	297769.1
14	1995	271955.8	304958.7	302224.7
15	1996	278575.1	305548.7	304958.7
16	1997	281151.7	309193.0	305548.7
17	1998	280228.3	311727.5	309193.0
18	1999	280262.4	311189.0	311727.5
19	2000	283412.8	310564.8	311189.0
20	2001	288066.6	306376.4	310564.8
21	2002	290377.5	309039.0	306376.4

## 2. TSPプログラムの作成

GiveWinのメニューバーにおいて、「File」 - 「New」 - 「Text」 とすればプログラムエディターが開く。そこに次のようにプログラムを入力していく。

freq a; smpl 1983 2002; read(file='h:ex3-3.xls'); olsq ce c yd yd1; ste=@res/@s; graph date ste; end;

プログラムを入力した後で、メニューバーから 「File」 - 「Save As」 を選び、ex3-3という 名前で保存する。ファイルの種類は TSP file (\*.TSP) を選ぶ。 プログラムの実行は「Modules」 - 「Run TSP」である。 3. ラグつき変数

この例 5 のモデルは CE = a + b YD + c YD-1 というように、説明変数に可処分所得の 1 期前の 値が含まれている。このような遅れ(ラグ)をともなう変数のことを、**ラグつき変数**という。 TSP では、プログラムの中でラグつき変数を計算することができる。

□ 手順

ex3-3.xls のファイルの D 列を削除し、ex3-3b.xls という名前で、Excel4.0 ワークシート 形式で保存する。 ex3-3.tsp のファイルを次のように修正し、「File」-「Save As」 を選び、ex3-3b という 名前で保存する。(変更箇所を太字で示した)

```
freq a;
smpl 1983 2002;
```

```
read(file='h:ex3-3b.xls');
print ce yd yd(-1);
olsq ce c yd yd(-1);
ste=@res/@s;
graph date ste;
end;
```

ラグつき変数を用いた場合、1983年の1期前のデータは存在しない。そのため、実行結果にお いていくつかの warning が出る。これを回避したい場合は、smpl 文で、データの期間を再度定 義しなおす必要がある。

#### 4. 相関係数、偏相関係数

TSP では 2 変量間の相関係数の行列を corr というコマンドで表示することができる。一方、 偏相関係数は計算して導出する必要がある。

ex3-3.tsp を次のように修正し、ex3-3c.tsp という名前で保存してみよう。なお、?で始まる文 はコメント行である。これらの行は TSP の操作に直接関係ないが、後で自分のプログラムを見て、 何をやったのかを確認する時にあると便利である。

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.xls');
?
? 重回帰分析
?
olsq ce c yd yd1;
ste=@res/@s;
graph date ste;
? 相関係数行列
?
corr ce yd yd1;
? 偏相関係数の計算
?
freq n;
smpl 1 1;
pcorr(1 = (@corr(1,2)-@corr(1,3)*@corr(2,3))/(sqrt(1-@corr(1,3)^2)*sqrt(1-@corr(2,3)^2));
pcorr13 = (@corr(1,3)-@corr(1,2)*@corr(2,3))/(sqrt(1-@corr(1,2)^2)*sqrt(1-@corr(2,3)^2));
pcorr23 = (@corr(2,3)-@corr(1,2)*@corr(1,3))/(sqrt(1-@corr(1,2)^2)*sqrt(1-@corr(1,3)^2));
?
? 偏相関係数の行列表示
?
pcorrdiag=1;
mmake pcorrmat1 pcorrdiag pcorr12 pcorr13;
mmake pcorrmat2 pcorr12 pcorrdiag pcorr23;
mmake pcorrmat3 pcorr13 pcorr23 pcorrdiag;
mmake(vert) pcorrmat pcorrmat1 pcorrmat2 pcorrmat3;
print pcorrmat;
end;
```

5. 外部ファイルへの書き出し

分析結果や、加工したデータファイルを外部ファイルへ書き出すことができる。Excel で入力 したデータを TSP で分析し、ふたたび Excel ファイルとして出力することも可能である。(これ によって両者の良い面が使える。)

例として、相関係数行列と偏相関係数行列を Excel 形式で書き出してみよう。end コマンドの前に次のようなコマンドを挿入してみよう。

?

? 相関係数行列、偏相関係数行列のファイル出力

- write(file='h:ex3-3corr.xls') @corr;
- write(file='h:ex3-3pcorr.xls') pcorrmat;

### <u>6. 分析結果の保存とWordへ読みこみ</u>

TSP の分析結果はテキストファイルとして保存できるので、それをエディターやワープロソフトを使って読みこむことができる。また、グラフも画像形式(デフォルトでは Enhanced MetaFile 形式)で保存できるので、Word などで読みこむことが可能である。

ここでは、ex3-3cの出力結果と残差プロットを Word ファイルにまとめてみよう。

🕮 手順

出力結果のウインドウをアクティブにして、「File」-「Save As」を選び、**ex3-3c**という 名前で保存する。(ここで、デフォルトのファイル名には、パス名がついているので、適当 に修正すること。)

残差プロットのウインドウをアクティブにして、「File」-「Save As」を選び、**残差プロ** ットという名前で保存する。

Word を開き、ex3-3c.out のファイルを呼び出す。ファイルの種類を「すべてのファイル」 にしておけば、すべての形式のファイルを呼び出すことが可能である。

グラフファイルは「挿入」-「図」-「ファイルから」で**残差プロット**のファイルを選べ ば良い。グラフの大きさなどは、右クリックで「図の書式設定」を選べば変更できる。 不要な部分を削除・修正し、タイトルと氏名を加えればレポートが完成する。